// File: contracts/ClientStoreFactory.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.17;
import {ClientTokenStore} from './ClientTokenStore.sol';
import {IClientTokenStoreFactory} from './interfaces/IClientTokenStoreFactory.sol';
import {Ownable} from 'oz/access/Ownable.sol';

contract ClientTokenStoreFactory is IClientTokenStoreFactory, Ownable {
  address public claimContract;

  constructor(address _claimContract) {
    claimContract = _claimContract;
  }

  function setClaimContract(address _claimContract) external onlyOwner {
    claimContract = _claimContract;
  }

  function createNewStore() external returns (address) {
    ClientTokenStore newStore = new ClientTokenStore(claimContract);
    newStore.transferOwnership(msg.sender);
    emit StoreCreated(address(newStore), msg.sender);
    return address(newStore);
  }
}

// File: contracts/ClientTokenStore.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.17;
import {IClientTokenStore} from './interfaces/IClientTokenStore.sol';
import {Ownable} from 'oz/access/Ownable.sol';
import {IERC20} from 'oz/token/ERC20/IERC20.sol';

contract ClientTokenStore is Ownable, IClientTokenStore {
  address public claimContract;

  constructor(address _claimContract) {
    claimContract = _claimContract;
  }

  function setClaimContract(address _claimContract) onlyOwner public {
    claimContract = _claimContract;
    //make event
  }

  function withdraw(address _token, uint _amount) external onlyOwner {
    IERC20 token = IERC20(_token);
    require(token.balanceOf(address(this)) >= _amount);
    token.transfer(owner(), _amount);

    emit Withdraw(owner(), _token, _amount);
  }

  function withdrawToReceiver(address _receiver, address _token, uint _amount) external onlyClaimer {
    IERC20 token = IERC20(_token);
    require(_receiver != address(0), "Receiver address invalid");
    require(_token != address(0), "Token address invalid");
    require(_amount > 0, "Amount must be greater than 0");
    require(token.balanceOf(address(this)) >= _amount, "Insufficient Tokens");
    token.transfer(_receiver, _amount);

    emit WithdrawToReceiver(_receiver, _token, _amount);
  }

  // requires allowance
  function deposit(address _token, uint _amount) external {
    require(_token != address(0), "Token address invalid");
    IERC20 token = IERC20(_token);
    require(
      token.allowance(msg.sender, address(this)) >= _amount,
      "Error: Insufficient Allowance"
    );
    token.transferFrom(msg.sender, address(this), _amount);
  }

  modifier onlyClaimer() {
    require(msg.sender == claimContract);
    _;
  }
}

// File: contracts/interfaces/IClientTokenStore.sol



//SPDX-License-Identifier: MIT
pragma solidity 0.8.17;
import {IERC20} from 'oz/token/ERC20/IERC20.sol';

interface IClientTokenStore {
  event Withdraw(address _to, address _token, uint _amount);
  event WithdrawToReceiver(address _receiver, address _token, uint _amount);
  function claimContract() external view returns (address);
  function withdraw(address _token, uint _amount) external;
  function withdrawToReceiver(address _receiver, address _token, uint _amount) external;
  function deposit(address _token, uint _amount) external;
}

// File: contracts/interfaces/IClientTokenStoreFactory.sol



//SPDX-License-Identifier: MIT
pragma solidity 0.8.17;

interface IClientTokenStoreFactory {
  event ClaimContractSet(address _claimContract);
  event StoreCreated(address _store, address _owner);
  function claimContract() external view returns (address);
  function createNewStore() external returns (address);
}

// File: oz/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: oz/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}

// File: oz/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}